Chapter 6

Floating Point

6.1 Floating Point Representation

6.1.1 Non-integral binary numbers

When number systems were discussed in the first chapter, only integer values were discussed. Obviously, it must be possible to represent non-integral numbers in other bases as well as decimal. In decimal, digits to the right of the decimal point have associated negative powers of ten:

\[ 0.123 = 1 \times 10^{-1} + 2 \times 10^{-2} + 3 \times 10^{-3} \]

Not surprisingly, binary numbers work similarly:

\[ 0.101_2 = 1 \times 2^{-1} + 0 \times 2^{-2} + 1 \times 2^{-3} = 0.625 \]

This idea can be combined with the integer methods of Chapter 1 to convert a general number:

\[ 110.011_2 = 4 + 2 + 0.25 + 0.125 = 6.375 \]

Converting from decimal to binary is not very difficult either. In general, divide the decimal number into two parts: integer and fraction. Convert the integer part to binary using the methods from Chapter 1. The fractional part is converted using the method described below.

Consider a binary fraction with the bits labeled \( a, b, c, \ldots \) The number in binary then looks like:

\[ 0.aabcdef \ldots \]

Multiply the number by two. The binary representation of the new number will be:

\[ a.bcdef \ldots \]
### CHAPTER 6. FLOOR AND CEILING

<table>
<thead>
<tr>
<th>Expression</th>
<th>Result</th>
<th>Bit Position</th>
</tr>
</thead>
<tbody>
<tr>
<td>0.5625 × 2</td>
<td>1.125</td>
<td>first bit = 1</td>
</tr>
<tr>
<td>0.125 × 2</td>
<td>0.25</td>
<td>second bit = 0</td>
</tr>
<tr>
<td>0.25 × 2</td>
<td>0.5</td>
<td>third bit = 0</td>
</tr>
<tr>
<td>0.5 × 2</td>
<td>1.0</td>
<td>fourth bit = 1</td>
</tr>
</tbody>
</table>

Figure 6.1: Converting 0.5625 to binary

<table>
<thead>
<tr>
<th>Expression</th>
<th>Result</th>
</tr>
</thead>
<tbody>
<tr>
<td>0.85 × 2</td>
<td>1.7</td>
</tr>
<tr>
<td>0.7 × 2</td>
<td>1.4</td>
</tr>
<tr>
<td>0.4 × 2</td>
<td>0.8</td>
</tr>
<tr>
<td>0.8 × 2</td>
<td>1.6</td>
</tr>
<tr>
<td>0.6 × 2</td>
<td>1.2</td>
</tr>
<tr>
<td>0.2 × 2</td>
<td>0.4</td>
</tr>
<tr>
<td>0.4 × 2</td>
<td>0.8</td>
</tr>
<tr>
<td>0.8 × 2</td>
<td>1.6</td>
</tr>
</tbody>
</table>

Figure 6.2: Converting 0.85 to binary

Note that the first bit is now in the one’s place. Replace the 0 with 0 to get:

\[ 0.bcdedef \ldots \]

and multiply by two again to get:

\[ b.cdeff \ldots \]

Now the second bit (b) is in the one’s position. This procedure can be repeated until as many bits needed are found. Figure 6.1 shows a real example that converts 0.5625 to binary. The method stops when a fractional part of zero is reached.

As another example, consider converting 23.85 to binary. It is easy to convert the integral part (23 = 10111₂), but what about the fractional part (0.85)? Figure 6.2 shows the beginning of this calculation. If one looks at
6.1. FLOATING POINT REPRESENTATION

the numbers carefully, an infinite loop is found! This means that 0.85 is a repeating binary (as opposed to a repeating decimal in base 10). There is a pattern to the numbers in the calculation. Looking at the pattern, one can see that $0.85 = 0.11011011_2$. Thus, $23.85 = 10111.11011011_2$.

One important consequence of the above calculation is that 23.85 cannot be represented exactly in binary using a finite number of bits. (Just as $\frac{1}{3}$ cannot be represented in decimal with a finite number of digits.) As this chapter shows, float and double variables in C are stored in binary. Thus, values like 23.85 cannot be stored exactly into these variables. Only an approximation of 23.85 can be stored.

To simplify the hardware, floating point numbers are stored in a consistent format. This format uses scientific notation (but in binary, using powers of two, not ten). For example, 23.85 or $10111.110110110\ldots_2$ would be stored as:

$$1.011111011001100110\ldots \times 2^{100}$$

(where the exponent (100) is in binary). A normalized floating point number has the form:

$$1.sssssssssssssss \times 2^{eeeeee}$$

where $1.sssssssssssssss$ is the significand and $eeeeee$ is the exponent.

6.1.2 IEEE floating point representation

The IEEE (Institute of Electrical and Electronic Engineers) is an international organization that has designed specific binary formats for storing floating point numbers. This format is used on most (but not all!) computers made today. Often it is supported by the hardware of the computer itself. For example, Intel’s numeric (or math) coprocessors (which are built into all its CPU’s since the Pentium) use it. The IEEE defines two different formats with different precisions: single and double precision. Single precision is used by float variables in C and double precision is used by double variables.

Intel’s math coprocessor also uses a third, higher precision called extended precision. In fact, all data in the coprocessor itself is in this precision. When it is stored in memory from the coprocessor it is converted to either single or double precision automatically. Extended precision uses a slightly different general format than the IEEE float and double formats and so will not be discussed here.

---

1. It should not be so surprising that a number might repeat in one base, but not another. Think about $\frac{1}{3}$, it repeats in decimal, but in ternary (base 3) it would be $0.1_3$.

2. Some compiler’s (such as Borland) long double type uses this extended precision. However, other compilers use double precision for both double and long double. (This is allowed by ANSI C.)
IEEE single precision

Single precision floating point uses 32 bits to encode the number. It is usually accurate to 7 significant decimal digits. Floating point numbers are stored in a much more complicated format than integers. Figure 6.3 shows the basic format of a IEEE single precision number. There are several quirks to the format. Floating point numbers do not use the two’s complement representation for negative numbers. They use a signed magnitude representation. Bit 31 determines the sign of the number as shown.

The binary exponent is not stored directly. Instead, the sum of the exponent and 7F is stored from bit 23 to 30. This biased exponent is always non-negative.

The fraction part assumes a normalized significand (in the form 1.sssssssss). Since the first bit is always an one, the leading one is not stored! This allows the storage of an additional bit at the end and so increases the precision slightly. This idea is know as the hidden one representation.

How would 23.85 be stored? First, it is positive so the sign bit is 0. Next the true exponent is 4, so the biased exponent is 7F + 4 = 83_{16}. Finally, the fraction is 01111011100110011001100 (remember the leading one is hidden). Putting this all together (to help clarify the different sections of the floating point format, the sign bit and the faction have been underlined and the bits have been grouped into 4-bit nibbles):

\[
0 \ 100 \ 0001 \ 1011 \ 1110 \ 1100 \ 1100 \ 1100_2 = 41BECCCC_{16}
\]

This is not exactly 23.85 (since it is a repeating binary). If one converts the above back to decimal, one finds that it is approximately 23.849998474. This number is very close to 23.85, but it is not exact. Actually, in C, 23.85 would not be represented exactly as above. Since the left-most bit that was truncated from the exact representation is 1, the last bit is rounded up to 1. So 23.85 would be represented as 41 BE CC CD in hex using single precision. Converting this to decimal results in 23.850000381 which is a slightly better approximation of 23.85.
6.1. FLOATING POINT REPRESENTATION

\[
\begin{array}{c|c|c|c|c|c}
\text{e} & \text{f} & \text{denotes the number zero} & \text{which can not be normalized} & \text{Note that there is a +0 and -0.} \\
0 & 0 & & & \\
0 & \neq 0 & \text{denotes a } \textit{denormalized number}. & \text{These are discussed in the next section.} \\
\text{FF} & 0 & \text{denotes infinity (} \infty \text{). } & \text{There are both positive} & \text{and negative infinities.} \\
\text{FF} & \neq 0 & \text{denotes an undefined result, known as } \textit{NaN} & \text{(Not a Number).} \\
\end{array}
\]

Table 6.1: Special values of \( f \) and \( e \)

<table>
<thead>
<tr>
<th>63</th>
<th>62</th>
<th>52</th>
<th>51</th>
<th>0</th>
</tr>
</thead>
<tbody>
<tr>
<td>s</td>
<td>e</td>
<td>f</td>
<td></td>
<td></td>
</tr>
</tbody>
</table>

Figure 6.4: IEEE double precision

How would -23.85 be represented? Just change the sign bit: C1 BE CC CD. Do not take the two’s complement!

Certain combinations of \( e \) and \( f \) have special meanings for IEEE floats. Table 6.1 describes these special values. An infinity is produced by an overflow or by division by zero. An undefined result is produced by an invalid operation such as trying to find the square root of a negative number, adding two infinities, etc.

Normalized single precision numbers can range in magnitude from \(1.0 \times 2^{-126} \approx 1.1755 \times 10^{-38}\) to \(1.1111 \ldots \times 2^{127} \approx 3.4028 \times 10^{38}\).

Denormalized numbers

Denormalized numbers can be used to represent numbers with magnitudes too small to normalize (\(i.e.\) below \(1.0 \times 2^{-126}\)). For example, consider the number \(1.001_2 \times 2^{-129} \approx 1.6530 \times 10^{-39}\). In the given normalized form, the exponent is too small. However, it can be represented in the unnormalized form: \(0.01001_2 \times 2^{-127}\). To store this number, the biased exponent is set to 0 (see Table 6.1) and the fraction is the complete significand of the number written as a product with \(2^{-127}\) (\(i.e.\) all bits are stored including the one to the left of the decimal point). The representation of \(1.001 \times 2^{-129}\) is then:

\[
0 \, 000 \, 0000 \, 0001 \, 0010 \, 0000 \, 0000 \, 0000 \, 0000
\]
IEEE double precision

IEEE double precision uses 64 bits to represent numbers and is usually accurate to about 15 significant decimal digits. As Figure 6.4 shows, the basic format is very similar to single precision. More bits are used for the biased exponent (11) and the fraction (52) than for single precision.

The larger range for the biased exponent has two consequences. The first is that it is calculated as the sum of the true exponent and 3FF (1023) (not 7F as for single precision). Secondly, a large range of true exponents (and thus a larger range of magnitudes) is allowed. Double precision magnitudes can range from approximately $10^{-308}$ to $10^{308}$.

It is the larger field of the fraction that is responsible for the increase in the number of significant digits for double values.

As an example, consider 23.85 again. The biased exponent will be $4 + 3FF = 403$ in hex. Thus, the double representation would be:

```
0 100 0000 0011 0111 1101 1001 1001 1001 1001 1001 1001 1001 1001 1010
```
or 40 37 D9 99 99 99 99 9A in hex. If one converts this back to decimal, one finds 23.850000000000014 (there are 12 zeros!) which is a much better approximation of 23.85.

The double precision has the same special values as single precision.\[5\]

Denormalized numbers are also very similar. The only main difference is that double denormalized numbers use $2^{-1023}$ instead of $2^{-127}$.

6.2 Floating Point Arithmetic

Floating point arithmetic on a computer is different than in continuous mathematics. In mathematics, all numbers can be considered exact. As shown in the previous section, on a computer many numbers can not be represented exactly with a finite number of bits. All calculations are performed with limited precision. In the examples of this section, numbers with an 8-bit significand will be used for simplicity.

6.2.1 Addition

To add two floating point numbers, the exponents must be equal. If they are not already equal, then they must be made equal by shifting the significand of the number with the smaller exponent. For example, consider $10.375 + 6.34375 = 16.71875$ or in binary:

```
1.0100110 \times 2^3
+ 1.1001011 \times 2^2
```

\[3\]The only difference is that for the infinity and undefined values, the biased exponent is 7FF not FF.
6.2. FLOATING POINT ARITHMETIC

These two numbers do not have the same exponent so shift the significand to make the exponents the same and then add:

\[
\begin{array}{c}
1.0100110 \times 2^3 \\
+ \quad 0.1100110 \times 2^3 \\
\hline
1.0000110 \times 2^4
\end{array}
\]

Note that the shifting of 1.1001011 \times 2^2 drops off the trailing one and after rounding results in 0.1100110 \times 2^3. The result of the addition, 10.0001100 \times 2^3 (or 1.00001100 \times 2^4) is equal to 10000.1102 or 16.75. This is not equal to the exact answer (16.71875)! It is only an approximation due to the round off errors of the addition process.

It is important to realize that floating point arithmetic on a computer (or calculator) is always an approximation. The laws of mathematics do not always work with floating point numbers on a computer. Mathematics assumes infinite precision which no computer can match. For example, mathematics teaches that \((a + b) - b = a\); however, this may not hold true exactly on a computer!

6.2.2 Subtraction

Subtraction works very similarly and has the same problems as addition. As an example, consider 16.75 - 15.9375 = 0.8125:

\[
\begin{array}{c}
1.0000110 \times 2^4 \\
- \quad 1.111111 \times 2^3 \\
\hline
0.000000 \times 2^4
\end{array}
\]

Shifting 1.111111 \times 2^3 gives (rounding up) 1.0000000 \times 2^4

\[
\begin{array}{c}
1.0000110 \times 2^4 \\
- \quad 1.000000 \times 2^4 \\
\hline
0.0000110 \times 2^4
\end{array}
\]

0.0000110 \times 2^4 = 0.11_2 = 0.75 which is not exactly correct.

6.2.3 Multiplication and division

For multiplication, the significands are multiplied and the exponents are added. Consider 10.375 \times 2.5 = 25.9375:

\[
\begin{array}{c}
1.0100110 \times 2^3 \\
\times \quad 1.0100000 \times 2^1 \\
\hline
10100110 \\
+ \quad 10100110 \\
\hline
1.10011111000000 \times 2^4
\end{array}
\]
Of course, the real result would be rounded to 8-bits to give:

\[ 1.1010000 \times 2^4 = 11010.000_2 = 26 \]

Division is more complicated, but has similar problems with round off errors.

### 6.2.4 Ramifications for programming

The main point of this section is that floating point calculations are not exact. The programmer needs to be aware of this. A common mistake that programmers make with floating point numbers is to compare them assuming that a calculation is exact. For example, consider a function named \( f(x) \) that makes a complex calculation and a program is trying to find the function’s root. One might be tempted to use the following statement to check to see if \( x \) is a root:

```c
if ( f(x) == 0.0 )
```

But, what if \( f(x) \) returns \( 1 \times 10^{-30} \)? This very likely means that \( x \) is a very good approximation of a true root; however, the equality will be false. There may not be any IEEE floating point value of \( x \) that returns exactly zero, due to round off errors in \( f(x) \).

A much better method would be to use:

```c
if ( fabs(f(x)) < EPS )
```

where \( EPS \) is a macro defined to be a very small positive value (like \( 1 \times 10^{-10} \)). This is true whenever \( f(x) \) is very close to zero. In general, to compare a floating point value (say \( x \)) to another (\( y \)) use:

```c
if ( fabs(x - y)/fabs(y) < EPS )
```

### 6.3 The Numeric Coprocessor

#### 6.3.1 Hardware

The earliest Intel processors had no hardware support for floating point operations. This does not mean that they could not perform float operations. It just means that they had to be performed by procedures composed of many non-floating point instructions. For these early systems, Intel did provide an additional chip called a *math coprocessor*. A math coprocessor has machine instructions that perform many floating point operations much faster than using a software procedure (on early processors, at least 10 times

---

\[ ^4 \text{A root of a function is a value } x \text{ such that } f(x) = 0 \]
faster!). The coprocessor for the 8086/8088 was called the 8087. For the 80286, there was a 80287 and for the 80386, a 80387. The 80486DX processor integrated the math coprocessor into the 80486 itself. Since the Pentium, all generations of 80x86 processors have a built-in math coprocessor; however, it is still programmed as if it was a separate unit. Even earlier systems without a coprocessor can install software that emulates a math coprocessor. These emulator packages are automatically activated when a program executes a coprocessor instruction and run a software procedure that produces the same result as the coprocessor would have (though much slower, of course).

The numeric coprocessor has eight floating point registers. Each register holds 80 bits of data. Floating point numbers are always stored as 80-bit extended precision numbers in these registers. The registers are named ST0, ST1, ST2, ... ST7. The floating point registers are used differently than the integer registers of the main CPU. The floating point registers are organized as a stack. Recall that a stack is a Last-In First-Out (LIFO) list. ST0 always refers to the value at the top of the stack. All new numbers are added to the top of the stack. Existing numbers are pushed down on the stack to make room for the new number.

There is also a status register in the numeric coprocessor. It has several flags. Only the 4 flags used for comparisons will be covered: C0, C1, C2 and C3. The use of these is discussed later.

### 6.3.2 Instructions

To make it easy to distinguish the normal CPU instructions from coprocessor ones, all the coprocessor mnemonics start with an F.

**Loading and storing**

There are several instructions that load data onto the top of the coprocessor register stack:

- **FLD source** loads a floating point number from memory onto the top of the stack. The source may be a single, double or extended precision number or a coprocessor register.
- **FILD source** reads an integer from memory, converts it to floating point and stores the result on top of the stack. The source may be either a word, double word or quad word.
- **FLD1** stores a one on the top of the stack.
- **FLDZ** stores a zero on the top of the stack.

There are also several instructions that store data from the stack into memory. Some of these instructions also pop (i.e. remove) the number from

---

5However, the 80486SX did not have an integrated coprocessor. There was a separate 80487SX chip for these machines.
the stack as it stores it.

- **FST dest** stores the top of the stack (ST0) into memory. The *destination* may either be a single or double precision number or a coprocessor register.

- **FSTP dest** stores the top of the stack into memory just as FST; however, after the number is stored, its value is popped from the stack. The *destination* may either a single, double or extended precision number or a coprocessor register.

- **FIST dest** stores the value of the top of the stack converted to an integer into memory. The *destination* may either a word or a double word. The stack itself is unchanged. How the floating point number is converted to an integer depends on some bits in the coprocessor’s *control word*. This is a special (non-floating point) word register that controls how the coprocessor works. By default, the control word is initialized so that it rounds to the nearest integer when it converts to integer. However, the **FSTCW** (Store Control Word) and **FLDCW** (Load Control Word) instructions can be used to change this behavior.

- **FISTP dest** Same as FIST except for two things. The top of the stack is popped and the *destination* may also be a quad word.

There are two other instructions that can move or remove data on the stack itself.

- **FXCH STn** exchanges the values in ST0 and STn on the stack (where n is register number from 1 to 7).

- **FFREE STn** frees up a register on the stack by marking the register as unused or empty.

### Addition and subtraction

Each of the addition instructions compute the sum of ST0 and another operand. The result is always stored in a coprocessor register.

- **FADD src** ST0 += src. The *src* may be any coprocessor register or a single or double precision number in memory.

- **FADD dest, ST0** dest += ST0. The *dest* may be any coprocessor register.

- **FADDP dest or FADDP dest, ST0** dest += ST0 then pop stack. The *dest* may be any coprocessor register.

- **FIADD src** ST0 += (float) src. Adds an integer to ST0. The *src* must be a word or double word in memory.

There are twice as many subtraction instructions than addition because the order of the operands is important for subtraction (*i.e.* \( a + b = b + a \), but \( a - b \neq b - a \)). For each instruction, there is an alternate one that subtracts in the reverse order. These reverse instructions all end in either
6.3. THE NUMERIC COPROCESSOR

```plaintext
segment .bss
array resq SIZE
sum resq 1

segment .text
    mov ecx, SIZE
    mov esi, array
    fldz ; ST0 = 0
lp:
    fadd qword [esi] ; ST0 += *(esi)
    add esi, 8 ; move to next double
    loop lp
    fstp qword sum ; store result into sum

Figure 6.5: Array sum example

R or RP. Figure 6.5 shows a short code snippet that adds up the elements of an array of doubles. On lines 10 and 13, one must specify the size of the memory operand. Otherwise the assembler would not know whether the memory operand was a float (dword) or a double (qword).

FSUB src ST0 -= src. The src may be any coprocessor register or a single or double precision number in memory.
FSUBR src ST0 = src - ST0. The src may be any coprocessor register or a single or double precision number in memory.
FSUB dest, ST0 dest -= ST0. The dest may be any coprocessor register.
FSUBR dest, ST0 dest = ST0 - dest. The dest may be any coprocessor register.
FSUBP dest or FSUBP dest, ST0 dest -= ST0 then pop stack. The dest may be any coprocessor register.
FSUBRP dest or FSUBRP dest, ST0 dest = ST0 - dest then pop stack. The dest may be any coprocessor register.
FISUB src ST0 -= (float) src. Subtracts an integer from ST0. The src must be a word or double word in memory.
FISUBR src ST0 = (float) src - ST0. Subtracts ST0 from an integer. The src must be a word or double word in memory.`
CHAPTER 6. FLOATING POINT

Multiplication and division

The multiplication instructions are completely analogous to the addition instructions.
- **FMUL src**
  
  ST0 *= src. The src may be any coprocessor register or a single or double precision number in memory.
- **FMUL dest, ST0**
  
  dest *= ST0. The dest may be any coprocessor register.
- **FMULP dest or FMULP dest, ST0**
  
  dest *= ST0 then pop stack. The dest may be any coprocessor register.
- **FIMUL src**
  
  ST0 *= (float) src. Multiplies an integer to ST0. The src must be a word or double word in memory.

Not surprisingly, the division instructions are analogous to the subtraction instructions. Division by zero results in an infinity.
- **FDIV src**
  
  ST0 /= src. The src may be any coprocessor register or a single or double precision number in memory.
- **FDIVR src**
  
  ST0 = src / ST0. The src may be any coprocessor register or a single or double precision number in memory.
- **FDIV dest, ST0**
  
  dest /= ST0. The dest may be any coprocessor register.
- **FDIVR dest, ST0**
  
  dest = ST0 / dest. The dest may be any coprocessor register.
- **FDIVP dest or FDIVP dest, ST0**
  
  dest /= ST0 then pop stack. The dest may be any coprocessor register.
- **FDIVRP dest or FDIVRP dest, ST0**
  
  dest = ST0 / dest then pop stack. The dest may be any coprocessor register.
- **FIDIV src**
  
  ST0 /= (float) src. Divides ST0 by an integer. The src must be a word or double word in memory.
- **FIDIVR src**
  
  ST0 = (float) src / ST0. Divides an integer by ST0. The src must be a word or double word in memory.

Comparisons

The coprocessor also performs comparisons of floating point numbers. The FCOM family of instructions does this operation.
6.3. THE NUMERIC COPROCESSOR

```plaintext
; if ( x > y )
;
fld qword [x] ; ST0 = x
fcomp qword [y] ; compare ST0 and y
fistw ax ; move C bits into FLAGS
sahf
jna else_part ; if x not above y, goto else_part
then_part:
; code for then part
jmp end_if
else_part:
; code for else part
end_if:
```

Figure 6.6: Comparison example

<table>
<thead>
<tr>
<th>Instruction</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>FCOM src</td>
<td>compares ST0 and src. The src can be a coprocessor register or a float or double in memory.</td>
</tr>
<tr>
<td>FCOMP src</td>
<td>compares ST0 and src, then pops stack. The src can be a coprocessor register or a float or double in memory.</td>
</tr>
<tr>
<td>FCOMPP</td>
<td>compares ST0 and ST1, then pops stack twice.</td>
</tr>
<tr>
<td>FICOM src</td>
<td>compares ST0 and (float) src. The src can be a word or dword integer in memory.</td>
</tr>
<tr>
<td>FICOMP src</td>
<td>compares ST0 and (float)src, then pops stack. The src can be a word or dword integer in memory.</td>
</tr>
<tr>
<td>FTST</td>
<td>compares ST0 and 0.</td>
</tr>
</tbody>
</table>

These instructions change the C₀, C₁, C₂ and C₃ bits of the coprocessor status register. Unfortunately, it is not possible for the CPU to access these bits directly. The conditional branch instructions use the FLAGS register, not the coprocessor status register. However, it is relatively simple to transfer the bits of the status word into the corresponding bits of the FLAGS register using some new instructions:

<table>
<thead>
<tr>
<th>Instruction</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>FSTSW dest</td>
<td>Stores the coprocessor status word into either a word in memory or the AX register.</td>
</tr>
<tr>
<td>SAHF</td>
<td>Stores the AH register into the FLAGS register.</td>
</tr>
<tr>
<td>LAHF</td>
<td>Loads the AH register with the bits of the FLAGS register.</td>
</tr>
</tbody>
</table>

Figure 6.6 shows a short example code snippet. Lines 5 and 6 transfer the C₀, C₁, C₂ and C₃ bits of the coprocessor status word into the FLAGS register. The bits are transferred so that they are analogous to the result of a comparison of two unsigned integers. This is why line 7 uses a JNA instruction.
The Pentium Pro (and later processors (Pentium II and III)) support two new comparison operators that directly modify the CPU’s FLAGS register.

**FCOMI src** compares ST0 and src. The src must be a coprocessor register.

**FCOMIP src** compares ST0 and src, then pops stack. The src must be a coprocessor register.

Figure 6.7 shows an example subroutine that finds the maximum of two doubles using the FCOMIP instruction. Do not confuse these instructions with the integer comparison functions (FICOM and FICOMP).

**Miscellaneous instructions**

This section covers some other miscellaneous instructions that the coprocessor provides.

- **FCHS ST0 = − ST0** Changes the sign of ST0
- **FABS ST0 = |ST0|** Takes the absolute value of ST0
- **FSQRT ST0 = √ST0** Takes the square root of ST0
- **FScale ST0 = ST0 × 2^{ST1}** multiples ST0 by a power of 2 quickly. ST1 is not removed from the coprocessor stack. Figure 6.8 shows an example of how to use this instruction.

### 6.3.3 Examples

#### 6.3.4 Quadratic formula

The first example shows how the quadratic formula can be encoded in assembly. Recall that the quadratic formula computes the solutions to the quadratic equation:

\[ ax^2 + bx + c = 0 \]

The formula itself gives two solutions for \( x \): \( x_1 \) and \( x_2 \).

\[ x_1, x_2 = \frac{-b \pm \sqrt{b^2 - 4ac}}{2a} \]

The expression inside the square root \((b^2 - 4ac)\) is called the *discriminant*. Its value is useful in determining which of the following three possibilities are true for the solutions.

1. There is only one real degenerate solution. \( b^2 - 4ac = 0 \)
2. There are two real solutions. \( b^2 - 4ac > 0 \)
3. There are two complex solutions. \( b^2 - 4ac < 0 \)

Here is a small C program that uses the assembly subroutine:
6.3. THE NUMERIC COPROCESSOR

______________________________  quadt.c  ________________________________

#include <stdio.h>

int quadratic ( double, double, double, double *, double *);

int main()
{

double a,b,c, root1, root2;

printf("Enter a, b, c :");
scanf("%lf %lf %lf", &a, &b, &c);
if ( quadratic ( a , b, c, &root1, &root2) )
printf("roots: %.10g %.10g
", root1, root2);
else
printf("No real roots
");
return 0;
}

______________________________  quadt.c  ________________________________

Here is the assembly routine:

______________________________  quad.asm  ________________________________

; function quadratic
; finds solutions to the quadratic equation:
; a*x^2 + b*x + c = 0
; C prototype:
; int quadratic( double a, double b, double c,
; double * root1, double *root2 )
; Parameters:
; a, b, c - coefficients of powers of quadratic equation (see above)
; root1 - pointer to double to store first root in
; root2 - pointer to double to store second root in
; Return value:
; returns 1 if real roots found, else 0

%define a  qword [ebp+8]
%define b  qword [ebp+16]
%define c  qword [ebp+24]
%define root1 dword [ebp+32]
%define root2 dword [ebp+36]
%define disc qword [ebp-8]
%define one_over_2a qword [ebp-16]

segment .data
MinuFour dw -4

segment .text
    global _quadratic
    _quadratic:
        push ebp
        mov ebp, esp
        sub esp, 16 ; allocate 2 doubles (disc & one_over_2a)
        push ebx ; must save original ebx

        fild word [MinusFour]; stack -4
        fld a ; stack: a, -4
        fld c ; stack: c, a, -4
        fmulp st1 ; stack: a*c, -4
        fmulp st1 ; stack: -4*a*c
        fld b
        fld b ; stack: b, b, -4*a*c
        fmulp st1 ; stack: b*b, -4*a*c
        faddp st1 ; stack: b*b - 4*a*c
        ftst ; test with 0
        fstsw ax
        sahf
        jb no_real_solutions ; if disc < 0, no real solutions
        fsqrt ; stack: sqrt(b*b - 4*a*c)
        fstp disc ; store and pop stack
        fld1 ; stack: 1.0
        fld a ; stack: a, 1.0
        fscale ; stack: a * 2^-1.0 = 2*a, 1
        fdivp st1 ; stack: 1/(2*a)
        fst one_over_2a ; stack: 1/(2*a)
        fld b ; stack: b, 1/(2*a)
        fld disc ; stack: disc, b, 1/(2*a)
        fsubrp st1 ; stack: disc - b, 1/(2*a)
        fmulp st1 ; stack: (-b + disc)/(2*a)
        mov ebx, root1
        fstp qword [ebx] ; store in *root1
        fld b ; stack: b
        fld disc ; stack: disc, b
        fchs ; stack: -disc, b
6.3. THE NUMERIC COPROCESSOR

6.3.5 Reading array from file

In this example, an assembly routine reads doubles from a file. Here is a short C test program:

```c
/*
 * This program tests the 32-bit read_doubles() assembly procedure.
 * It reads the doubles from stdin. (Use redirection to read from file.)
 */
#include <stdio.h>
extern int read_doubles( FILE *, double *, int );
#define MAX 100

int main()
{
    int i, n;
    double a[MAX];
    n = read_doubles(stdin, a, MAX);
    for ( i=0; i < n; i++ )
        printf(“%3d %g\n”, i, a[i]);
    return 0;
}
```
Chapter 6. Floating Point

Here is the assembly routine

```assembly
segment .data
format db "%lf", 0 ; format for fscanf()

segment .text
global _read_doubles
extern _fscanf

%define SIZEOF_DOUBLE  8
%define FP dword [ebp + 8]
%define ARRAYP dword [ebp + 12]
%define ARRAY_SIZE dword [ebp + 16]
%define TEMP_DOUBLE [ebp - 8]

; function _read_doubles
; C prototype:
; int read_doubles( FILE * fp, double * arrayp, int array_size );
; This function reads doubles from a text file into an array, until
; EOF or array is full.
; Parameters:
; fp - FILE pointer to read from (must be open for input)
; arrayp - pointer to double array to read into
; array_size - number of elements in array
; Return value:
; number of doubles stored into array (in EAX)

_read_doubles:
    push ebp
    mov ebp,esp
    sub esp, SIZEOF_DOUBLE ; define one double on stack

    push esi ; save esi
    mov esi, ARRAYP ; esi = ARRAYP
    xor edx, edx ; edx = array index (initially 0)

    while_loop:
        cmp edx, ARRAY_SIZE ; is edx < ARRAY_SIZE?
        jnl short quit ; if not, quit loop
```

6.3. THE NUMERIC COPROCESSOR

6.3.6 Finding primes

This final example looks at finding prime numbers again. This implementation is more efficient than the previous one. It stores the primes it has found in an array and only divides by the previous primes it has found instead of every odd number to find new primes.
One other difference is that it computes the square root of the guess for the next prime to determine at what point it can stop searching for factors. It alters the coprocessor control word so that when it stores the square root as an integer, it truncates instead of rounding. This is controlled by bits 10 and 11 of the control word. These bits are called the RC (Rounding Control) bits. If they are both 0 (the default), the coprocessor rounds when converting to integer. If they are both 1, the coprocessor truncates integer conversions. Notice that the routine is careful to save the original control word and restore it before it returns.

Here is the C driver program:

```
#include <stdio.h>
#include <stdlib.h>

/*
 * function find_primes
 * finds the indicated number of primes
 * Parameters:
 *  a — array to hold primes
 *  n — how many primes to find
 */
eextern void find_primes ( int *a , unsigned n );

int main()
{
    int status ;
    unsigned i ;
    unsigned max ;
    int *a ;

    printf ( "How many primes do you wish to find? " ) ;
    scanf ( "%u" , &max ) ;
    a = calloc ( sizeof(int) , max ) ;

    if ( a ) {
        find_primes (a, max) ;
        /* print out the last 20 primes found */
        for ( i = ( max > 20 ) ? max - 20 : 0 ; i < max ; i++ )
            printf ( "%3d %d\n" , i+1 , a[i]) ;
    }
    return status ;
}
```
6.3. THE NUMERIC COPROCESSOR

```c
free(a);
status = 0;
}
else {
    fprintf(stderr, "Can not create array of %u ints\n", max);
    status = 1;
}

return status;
```

Here is the assembly routine:

```
segment .text
    global _find_primes
;
; function find_primes
; finds the indicated number of primes
; Parameters:
;    array - array to hold primes
;    n_find - how many primes to find
; C Prototype:
;extern void find_primes( int * array, unsigned n_find )
;
%define array ebp + 8
%define n_find ebp + 12
%define n ebp - 4 ; number of primes found so far
%define isqrt ebp - 8 ; floor of sqrt of guess
%define orig_cntl_wd ebp - 10 ; original control word
%define new_cntl_wd ebp - 12 ; new control word

_find_primes:
    enter 12,0 ; make room for local variables
    push ebx ; save possible register variables
    push esi
    fstcw word [orig_cntl_wd] ; get current control word
    mov ax, [orig_cntl_wd]
```

or    ax, OC00h          ; set rounding bits to 11 (truncate)
mov   [new_cntl_wd], ax
fldcw  word [new_cntl_wd]

mov esi, [array]          ; esi points to array
mov dword [esi], 2        ; array[0] = 2
mov dword [esi + 4], 3    ; array[1] = 3
mov ebx, 5                ; ebx = guess = 5
mov dword [n], 2          ; n = 2

; This outer loop finds a new prime each iteration, which it adds to the
; end of the array. Unlike the earlier prime finding program, this function
; does not determine primeness by dividing by all odd numbers. It only
; divides by the prime numbers that it has already found. (That’s why they
; are stored in the array.)

while_limit:
  mov    eax, [n]
  cmp    eax, [n_find]      ; while ( n < n_find )
  jnb    short quit_limit

  mov    ecx, 1            ; ecx is used as array index
  push   ebx                ; store guess on stack
  fild   dword [esp]        ; load guess onto coprocessor stack
  pop    ebx                ; get guess off stack
  fsqrt
  fistp  dword [isqrt]      ; isqrt = floor(sqrt(guess))

; This inner loop divides guess (ebx) by earlier computed prime numbers
; until it finds a prime factor of guess (which means guess is not prime)
; or until the prime number to divide is greater than floor(sqrt(guess))

while_factor:
  mov    eax, dword [esi + 4*ecx]     ; eax = array[ecx]
  cmp    eax, [isqrt]                  ; while ( isqrt < array[ecx]
  jnbe   short quit_factor_prime
  mov    eax, ebx
  xor    edx, edx
  div    dword [esi + 4*ecx]
  or     edx, edx
  jz    short quit_factor_not_prime
  inc    ecx                        ; try next prime
6.3. THE NUMERIC COPROCESSOR

```assembly
69   jmp short while_factor
70
71 ; found a new prime!
72 ;
73 quit_factor_prime:
74   mov eax, [n]
75   mov dword [esi + 4*eax], ebx  ; add guess to end of array
76   inc eax
77   mov [n], eax                  ; inc n
78
79 quit_factor_not_prime:
80   add ebx, 2                    ; try next odd number
81   jmp short while_limit
82
83 quit_limit:
84
85   fldcw word [orig_cntl_wd]     ; restore control word
86   pop esi                      ; restore register variables
87   pop ebx
88
89 leave
90 ret
```

_________________________ prime2.asm ___________________________
global _dmax

segment .text
; function _dmax
; returns the larger of its two double arguments
; C prototype
; double dmax( double d1, double d2 )
; Parameters:
; d1 - first double
; d2 - second double
; Return value:
; larger of d1 and d2 (in ST0)
%define d1 ebp+8
%define d2 ebp+16
_dmax:
    enter 0, 0
    fld qword [d2]
    fld qword [d1] ; ST0 = d1, ST1 = d2
    fcomip st1 ; ST0 = d2
    jna short d2_bigger
    fcomp st0 ; pop d2 from stack
    fld qword [d1] ; ST0 = d1
    jmp short exit

 d2_bigger: ; if d2 is max, nothing to do
    exit:
    leave
    ret

Figure 6.7: FCOMIP example
6.3. THE NUMERIC COPROCESSOR

Figure 6.8: FSCALE example

```
segment .data
x dq 2.75 ; converted to double format
five dw 5

segment .text
fld dword [five] ; ST0 = 5
fld qword [x] ; ST0 = 2.75, ST1 = 5
fscale ; ST0 = 2.75 * 32, ST1 = 5
```